
Problem description: 
 
This example represents the propagation of a seismic wave in a sediment valley. The 
wave is generated at the centre of the region, while a “receiver” (displacement 
transducer) is located at a certain height over the source. 
The bedrock, containing the source, is modelled by spectral elements, while the 
sediment valley may either be meshed by finite elements (hybrid solution), or by 
spectral elements (purely spectral solution). 
By comparing the two solutions, an assessment of the quality of FE/SE coupling is 
obtained. 
 
Numerical Solutions 
 
VALL02 
 
This calculation assumes a hybrid mesh composed of  320 macro spectral elements, 
5120 micro spectral elements and 3035 finite elements. The coupling between FE and 
SE regions along the closed square interface is realized by means of the LINK COUP 
FESE directive. 
 
The mesh generation file is: 
 
*%siz 50 
* 
opti echo 0; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxleg2.proc'; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxspect2.proc'; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxpdroi1.proc'; 
opti echo 1; 
* 
opti dime 2 elem qua4; 
opti titr 'VALL - 02'; 
* 
p0=0 0; 
p1=500 0; 
p2=0 300; 

p3=500 450; 
p4=0 500; 
p5=500 500; 
n=4; 
tol=0.01; 
tol2=0.1; 
* 
i=0; repe lab1 20; 
  i=i + 1; 
* 
  xi = 25. * i; 
  yy = cos( xi * 180. / 500.); 
  yi = 375. - (75. * yy); 
  pti = xi yi; 

  si (ega i 1); 
    c1 = d 1 p2 pti; 
  sinon; 
    c1 = c1 d 1 pti; 
  finsi; 
* 
fin lab1; 
elim tol (c1 et p3); 
* 
c2 = p3 d 16 p1; 
c3 = p1 d 20 p0; 
c4 = p0 d 16 p2; 
* 
sm = dall c1 c2 c3 c4 plan; 
* 
sem = pxspect2 sm n tol; 
* 
p2p=p2 plus p0; 
p3p=p3 plus p0; 
* 
i=0; repe lab2 20; 
  i=i + 1; 
* 
  xi = 25. * i; 
  yy = cos( xi * 180. / 500.); 
  yi = 375. - (75. * yy); 
  pti = xi yi; 
  si (ega i 1); 
    c1p = d 4 p2p pti; 
  sinon; 
    c1p = c1p d 4 pti; 
  finsi; 
* 
fin lab2; 
elim tol (c1p et p3p); 
* 
c2p = p3p d 5 p5; 
c3p = p5 d 80 p4; 

c4p = p4 d 20 p2p; 
opti elem tri3; 
fem = surf (c1p et c2p et c3p et c4p) plan; 
* 
ptiold=p2; 
i=0; repe lab3 20; 
  i=i + 1; 
* 
  xi = 25. * i; 
  yy = cos( xi * 180. / 500.); 
  yi = 375. - (75. * yy); 
  pti = xi yi; 
  si (i ega 1); 
    nods = pxpdroi1 sem ptiold pti tol2; 
  sinon; 
    nods = nods et (pxpdroi1 sem ptiold pti tol2); 
  finsi; 
  ptiold = pti; 
* 
fin lab3; 
nodf=c1p; 
* 
tpln=0.0 400.0; 
elim tol (tpln et fem); 
* 
blxs = pxpdroi1 sem p0 p2 tol2; 
blxf = c4p; 
* 
trac (fem et sem); 
* 
mesh = fem et sem et sm et nodf et nods et tpln et blxs et blxf; 
* 
tass mesh; 
* 
opti sauv form 'vall02.msh'; 
sauv form mesh; 
fin; 

 
The input file is: 
VALL - 02 
$ 
ECHO 
$VERI 
!CONV win 
CAST MESH 
DPLA NONL 
$ 
DIME 
  PT2L 6876 MS24 320 S24 5120 TRIA 3035 ZONE 3 
TERM 
$ 
GEOM MS24 sm S24 sem TRIA fem TERM 
$ 
MATE LINE RO 2500. YOUN 1.501196172E9 NU 0.200956938 
          LECT sm fem TERM 
$ 
LINK COUP 
     CONT SPLA NX 1.0 NY 0.0 LECT blxf blxs TERM 
     FESE FNOD LECT nodf TERM 
          SNOD LECT nods TERM 
CHAR SPEC POIN BET PRES 
               SOUR BETA 200.0 
                    AMP 10.0  
                    T0 0.01 
                    X 0.0 Y 200.0 
                    ALFA 20.0 

$ 
ECRI DEPL VITE ACCE FINT FEXT TFRE 0.5 
     FICH ALIC FREQ 5 
$ 
OPTI PAS UTIL NOTE 
 LOG 1 
$ 
CALC TINI 0. TEND 0.5d0 PASF 1.00D-3 
*================================================================= 
SUIT 
Post-treatment (bande alice) 
ECHO 
* 
RESU alic GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR 1 'dy_rec'  DEPL COMP 2 NOEU LECT tpln TERM 
* 
trac 1 AXES 1.0 'DISPL. [M]' yzer 
list 1 AXES 1.0 'DISPL. [M]' 
* 
QUAL DEPL COMP 2 LECT tpln TERM REFE  9.95286E-9 TOLE 3.E-3 
*================================================================= 
FIN 

 
VALLPS 
This calculation assumes a purely spectral mesh composed of 480 macro spectral 
elements and 7680 micro spectral elements. The mesh generation file is: 
 
*%siz 50 
* 
opti echo 0; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxleg2.proc'; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxspect2.proc'; 
opti donn 'D:\Users\Folco\Plexis3c\Proc\pxpdroi1.proc'; 
opti echo 1; 
* 
opti dime 2 elem qua4; 
opti titr 'VALL - PS'; 
* 
p0=0 0; 
p1=500 0; 
p2=0 300; 
p3=500 450; 
p4=0 500; 
p5=500 500; 
n=4; 
tol=0.01; 
tol2=0.1; 
* 
i=0; repe lab1 20; 
  i=i + 1; 
* 
  xi = 25. * i; 
  yy = cos( xi * 180. / 500.); 
  yi = 375. - (75. * yy); 
  pti = xi yi; 
  si (ega i 1); 
    c1 = d 1 p2 pti; 
  sinon; 
    c1 = c1 d 1 pti; 
  finsi; 
* 

fin lab1; 
elim tol (c1 et p3); 
* 
c2 = p3 d 16 p1; 
c3 = p1 d 20 p0; 
c4 = p0 d 16 p2; 
* 
sm1 = dall c1 c2 c3 c4 plan; 
* 
c5 = p3 d 8 p5; 
c6 = p5 d 20 p4; 
c7 = p4 d 8 p2; 
* 
sm2 = dall c1 c5 c6 c7 plan; 
* 
trac (sm1 et sm2); 
* 
sm = sm1 et sm2; 
* 
sem = pxspect2 sm n tol; 
* 
tpln=0.0 400.0; 
elim tol (tpln et sem); 
* 
blxs = pxpdroi1 sem p0 p4 tol2; 
* 
mesh = sem et sm et tpln et blxs ; 
* 
tass mesh; 
* 
opti sauv form 'vallPS.msh'; 
sauv form mesh; 
fin; 
 

The input file is: 
 
VALL - PS 
$ 
ECHO 
$VERI 
!CONV win 
CAST MESH 
DPLA NONL 
$ 
DIME 
  PT2L 7857 MS24 480 S24 7680 ZONE 2 
TERM 
$ 
GEOM MS24 sm S24 sem TERM 
$ 
MATE LINE RO 2500. YOUN 1.501196172E9 NU 0.200956938 
          LECT sm TERM 
$ 
LINK COUP 
     CONT SPLA NX 1.0 NY 0.0 LECT blxs TERM 
* 
CHAR SPEC POIN BET PRES 
               SOUR BETA 200.0 
                    AMP 10.0  
                    T0 0.01 
                    X 0.0 Y 200.0 
                    ALFA 20.0 
$ 

ECRI DEPL VITE ACCE FINT FEXT TFRE 0.5 
     FICH ALIC FREQ 10 
$ 
OPTI PAS UTIL NOTE 
$ 
CALC TINI 0. TEND 0.5d0 PASF 0.50D-3 
*================================================================= 
SUIT 
Post-treatment (bande alice) 
ECHO 
* 
RESU alic GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR 1 'dy_rec_ps' DEPL COMP 2 NOEU LECT tpln TERM 
RCOU 2 'dy_rec' FICH 'vall02.pun' 
* 
trac 1 AXES 1.0 'DISPL. [M]' yzer 
trac 2 1 AXES 1.0 'DISPL. [M]' yzer 
         COLO noir roug 
* 
QUAL DEPL COMP 2 LECT tpln TERM REFE  1.07909E-8 TOLE 3.E-3 
*================================================================= 
FIN 

The receiver signal in the coupled FE/SE case is: 
 

 
 
The receiver signal in the purely spectral case is: 

The final displacement norm in the hybrid solution is: 
 

 
 
The final displacement norm in the purely spectral solution is: 
 

 
 


